# $Id: PKGBUILD 85519 2013-03-02 15:17:37Z arodseth $
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Maato <maato softwarebakery com>

pkgname=volumeicon
pkgver=0.5.0
pkgrel=1
pkgdesc='Volume control for your system tray'
arch=('x86_64' 'i686')
url='http://softwarebakery.com/maato/volumeicon.html'
license=('GPL3')
depends=('gtk3' 'alsa-lib' 'libnotify')
makedepends=('intltool')
source=("http://softwarebakery.com/maato/files/volumeicon/$pkgname-$pkgver.tar.gz")
sha256sums=('e5f1179f9ec5ab25e3740e4f6bbe0baa75368f0ae87d370197b3fbefb61bd782')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  sed -i 's/0.23/0.50.2/' configure.ac
  autoconf
  ./configure \
    --prefix=/usr \
    --enable-notify
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"

  make DESTDIR="$pkgdir/" install
}

# vim:set ts=2 sw=2 et:
